



Giving SAS Amnesia

A `%reset` macro

9/Feb/2005

Giving SAS Amnesia

- When developing a SAS program I often type a code segment, run it, fix my typos, run it again, and so on.
- What happens if I accidentally “break” an important `libname` or `format` statement?
- What happens if I “break” the code that produces a `WORK` dataset?

Giving SAS Amnesia

- SAS remembers libnames and formats until you (successfully) change them or exit and re-enter SAS.
- Similarly, old copies of WORK datasets remain.
- That can be very problematic for debugging.
- It's also very embarrassing giving a seriously broken program that you thought works to someone else for checking! ☹️



Before

```
1 data males;  
2     set sasuser.diabetes;  
3     DiffGluc=PostGluc-FastGluc;  
4     where sex="M";  
5 run;
```

NOTE: There were 9 observations read from the data set SASUSER.DIABETES.

```
WHERE sex='M';
```

NOTE: The data set WORK.MALES has 9 observations and 9 variables.

NOTE: DATA statement used:

real time	0.34 seconds
cpu time	0.10 seconds

Before, continued

```
6   proc means data=males;  
7       var DiffGluc;  
8   run;
```

NOTE: There were 9 observations read from the data set WORK.MALES.

NOTE: PROCEDURE MEANS used:

real time	0.46 seconds
cpu time	0.06 seconds

Results

The MEANS Procedure

Analysis Variable : DiffGluc

N	Mean	Std Dev	Minimum	Maximum
9	53.3333333	5.0000000	42.0000000	58.0000000

After

```
9      data males;
10         set sasuser.diabetes;
11         DiffGluc=PostGluc-FastGluc;
12         where sex="M"
```

NOTE: SCL source line.

```
13      run;
```

22

76

ERROR: Syntax error while parsing WHERE clause.

...

WARNING: Data set WORK.MALES was not replaced because this step was stopped.

NOTE: DATA statement used:

real time	0.06 seconds
-----------	--------------

cpu time	0.03 seconds
----------	--------------

After, continued

```
14   proc means data=males;  
15       var DiffGluc;  
16   run;
```

NOTE: There were 9 observations read from the data set WORK.MALES.

NOTE: PROCEDURE MEANS used:

real time	0.01 seconds
cpu time	0.01 seconds

Results

The MEANS Procedure

Analysis Variable : DiffGluc

N	Mean	Std Dev	Minimum	Maximum
9	53.3333333	5.0000000	42.0000000	58.0000000

%RESET

- Get rid of any junk datasets.
- Forget all librefs that might have been set.
- Delete the output to prevent multiple copies.

Empty WORK libref

- Remove all datasets from the WORK libref.
- Get rid of all formats that were not stored in a permanent libref.
- We can't blindly delete all the compiled macros (%reset is itself a macro).
Something more precise is left for a later version.

```
PROC DATASETS LIBRARY=WORK KILL  
MEMTYPE=DATA NODETAILS NOLIST;  
QUIT;
```

- “KILL” deletes everything in the specified library, subject to any MEMTYPE= specification.
- Leaving off MEMTYPE=DATA will make it try to delete everything, which will result in errors (at a minimum it will object to deleting the macros).
- “NODETAILS NOLIST” keeps PROC DATASETS quiet about what it is doing.

```
PROC DATASETS LIBRARY=WORK  
NODETAILS NOLIST;  
DELETE FORMATS /MEMTYPE=CATALOG;  
QUIT;
```

- Deletes all formats that have not been explicitly stored elsewhere.
- Will result in a NOTE: message if there have been no formats defined.

Clear all librefs

- In case there is an error pertaining to assigning librefs in our program, we need to clear any currently assigned ones.
- SASUSER, SASHELP, MAPS, and WORK can stay.

```
libname _all_ clear;
```

- Note that the “keyword” `_all_` in this particular case means all *except* the standard librefs.
- It won't clear SASUSER, WORK, etc.

Clear Previous Output

- **DM OUT "CLEAR" ;**
- This only affects default output (with ODS unspecified or set to LISTING).
- I've considered clearing the log (with DM LOG "CLEAR"), but have decided I prefer doing that manually.

Reset the page number

```
OPTIONS PAGENO=1;
```

Reset Headers and Footers

```
title;
```

```
footnote;
```

Note that this completely removes the titles and footnotes. If you want to go back to the default, you would need to change the title statement to

```
title1 "The SAS System";
```

Show only the log and editor windows

- Without the following statements, the (blank) output window or the old program editor window may be left in front, which may be a nuisance.
- **DM PGM "ICON" ;**
- **DM OUT "ICON" ;**

Putting it all together

```
%MACRO RESET;  
PROC DATASETS LIBRARY=WORK KILL  
MEMTYPE=DATA NODETAILS NOLIST;  
PROC DATASETS LIBRARY=WORK  
NODETAILS NOLIST;  
DELETE FORMATS /MEMTYPE=CATALOG;  
QUIT;  
DM OUT "CLEAR";
```

Putting it all together, continued

```
DM PGM "ICON";  
DM OUT "ICON";  
OPTIONS PAGENO=1;  
title;  
footnote;  
libname _all_ clear;  
%MEND RESET;
```

New to SAS 9.1!

```
options dmssynchk;
```

Program error without dmssynchk

```
10  options nodmssynchk; *DEFAULT;
11  data males;
12      set sasuser.diabetes;
13      DiffGluc=PostGluc-FastGluc;
14      where sex="M"
15  run;
    ---
    22
    76
```

ERROR: Syntax error while parsing WHERE clause.

...

WARNING: Data set WORK.MALES was not replaced because this step was stopped.

NOTE: DATA statement used (Total process time):

real time	0.18 seconds
cpu time	0.01 seconds

Without dmssynchk, continued

```
16   proc means data=males;  
17       var DiffGluc;  
18   run;
```

NOTE: There were 9 observations read from the data set WORK.MALES.

NOTE: PROCEDURE MEANS used (Total process time):

real time	0.01 seconds
cpu time	0.01 seconds

Program error with dmssynchk

```
19  options dmssynchk;
20  data males;
21      set sasuser.diabetes;
22      DiffGluc=PostGluc-FastGluc;
23      where sex="M"
24  run;
    ---
    22
    76
```

ERROR: Syntax error while parsing WHERE clause.

...

WARNING: Data set WORK.MALES was not replaced because this step was stopped.

NOTE: DATA statement used (Total process time):

real time	0.04 seconds
-----------	--------------

cpu time	0.03 seconds
----------	--------------

With dmssynchk, continued


```
25  proc means data=males;  
26      var DiffGluc;  
27  run;
```

NOTE: The SAS System stopped processing this step because of errors.

NOTE: PROCEDURE MEANS used (Total process time):

real time	0.01 seconds
cpu time	0.01 seconds

Questions?



Douglas Martin

Biomira Inc.

450-3761 ext. 137

dmartin@biomira.com

(for an electronic copy of this presentation, please e-mail me)